
;  **** IDL_2dtv_bar.pro *** Ling-Hsiao Lyu 04/22/2009 
; 
;====================================== 
fn00='Jz' 
;fn00='Ez' 
;fn00='By' 
; save,time,lx,ly,ax,ay,data,fn0f,filename=fn00+'2D.save' 
restore,fn00+'2D.save' 
 
myftitle='!8J!Dz!N/J!D0!N  time='+string(format='(f4.1)',time) 
;myftitle='!8E!Dz!N/J!D0!N  time='+string(format='(f4.1)',time) 
;myftitle='!8B!Dy!N/J!D0!N  time='+string(format='(f4.1)',time) 
 
;====================================== 
; Set up Device 
; 
;!P.FONT=0 
set_plot,'ps' 
; 
; device default length unit is centimeter 
; define page size to be A4 size: 21cm x 29 cm  
; one must define bits_per_pixel=8 
;  otherwise, default bits_per_pixel=4, result in poor gray level image 
;  *****The option xoffset=0.,yoffset=0.  is a necessary option*****(err1) 
; 
xpage=21 & ypage=29  
device,xsize=xpage,ysize=ypage, xoffset=0.,yoffset=0.,bits_per_pixel=8 
;,/centimeters ; this is the default device length unit 
;device,/color  ;if loadct,ict >0 
;device,/HELVETICA,/BOLD   ;if without !P.FONT=0, this statement is useless 
; 
fnps=fn00+'_IDL_2dtv_bar.ps' 
device,file=fnps 
myxtitle='!8x/x!D0!N' 
myytitle='!8y/x!D0!N' 
maintitle=myftitle 
;====================================== 
print,'device size in cm (xpage,ypage)=',xpage, ypage 
; 
; real size: for TVScl 
; 
wx=10. & wy=10.  ;pre-set wx and wy to be real number 
;print,'enter wx, wy (image size in cm, e.g., 10, 15)' 
;read,wx,wy 
; 
nbar=20 
wbar=wx/nbar 
; 
xedge=(xpage-wx-2*wbar)*0.5 
xbedge=xedge+wx+wbar 
yedge=(ypage-wy)*0.5 
; 
; normalized size: for plot 
; 
xl=wx/xpage & xorg=xedge/xpage 
yl=wy/ypage & yorg=yedge/ypage 
xlbar=wbar/xpage 
x0=xorg & x1=x0+xl & x2=x1+xlbar & x3=x2+xlbar 
y0=yorg & y1=y0+yl & y2=y0       & y3=y1 
; 



;====================================== 
;ict=0 
ict=33 
;print,'enter ict (e.g., 0 for gray level, or 33 for color level)' 
;read,ict 
if ict gt 0 then device,/color  ; for loadct,ict>0 
LoadCT,ict 
; 
ist_low=0 & ist_high=255 
;ist_low=255 & ist_high=0 
;read, ist_low, ist_high 
stretch,ist_low,ist_high 
;====================================== 
lxmin=1 & lxmax=lx & lymin=1 & lymax=ly 
;lxmin=? & lxmax=? & lymin=? & lymax=? 
;print,'enter lxmin, lxmax, lymin, lymax, (e.g., 1,',lx,'1,',ly,')' 
;read, lxmin, lxmax, lymin, lymax 
lxplot=lxmax-lxmin+1 & xmin=ax(lxmin-1) & xmax=ax(lxmax-1) 
lyplot=lymax-lymin+1 & ymin=ay(lymin-1) & ymax=ay(lymax-1) 
 
data2d=fltarr(lxplot,lyplot,/nozero) 
; build  data2d 
for ii=0,lxplot-1 do begin &$ 
for jj=0,lyplot-1 do begin &$ 
 i= ii+lxmin-1 & j=jj+lymin-1 & data2d(ii:ii,jj:jj)=data(i:i,j:j) &$ 
endfor &$ 
endfor 
; 
;====================================== 
; find fmin, fmax 
fmax=max(data2d) 
fmin=min(data2d) 
print,'fmin,fmax=',fmin,fmax 
; 
; *************Change fmin, fmax************* 
; 
;print,'enter new fmin_new,fmax_new'  
;read, fmin_new,fmax_new 
; 
; increase the range of (fmin,fmax): for fixed color level in animation 
; 
;if fmin_new lt fmin then data2d(0:0,0:0)=fmin_new 
;if fmax_new gt fmax then data2d(0:0,1:1)=fmax_new 
; 
; reduce the range of (fmin,fmax):  
;      to see the detail fine structures in the range of (fmin,fmax) 
; 
;if fmin_new gt fmin then begin &$ 
;for ii=0,lxplot-1 do begin &$ 
;for jj=0,lyplot-1 do begin &$ 
; if data2d(ii:ii,jj:jj) lt fmin_new then data2d(ii:ii,jj:jj)=fmin_new &$ 
;; if data2d(ii:ii,jj:jj) lt fmin_new then data2d(ii:ii,jj:jj)=fmax_new &$ 
;endfor &$ 
;endfor &$ 
;endif 
; 
;if fmax_new lt fmax then begin &$ 
;for ii=0,lxplot-1 do begin &$ 
;for jj=0,lyplot-1 do begin &$ 
; if data2d(ii:ii,jj:jj) gt fmax_new then data2d(ii:ii,jj:jj)=fmax_new &$ 



;; if data2d(ii:ii,jj:jj) gt fmax_new then data2d(ii:ii,jj:jj)=fmin_new &$ 
;endfor &$ 
;endfor &$ 
;endif 
;fmin=fmin_new & fmax=fmax_new  
;====================================== 
; build bar2d from fmax, fmin 
lxbar=2 
lybar=100 
bar2d=fltarr(lxbar,lybar,/nozero) 
delf=(fmax-fmin)/(lybar-1) 
for ii=0,lybar-1 do bar2d(*,ii:ii) = fmin+ii*delf  
;======================================================= 
; ***********BEGINNING OF PLOT********************** 
; make each plot twice 
; (plot the axes one more time after the TVSCAL)  
; 
!P.Multi=[0,4,1] ;[nth_plot,ncolumn, nrow]  
;!P.charsize=1.25  ;for one plot 
!P.charsize=2.5  ;for three plots 
;====================================== 
; Define ploting area: [lower-left (x,y), upper-right (x,y)] in normalize scale 
; 
; ****The following plot statement should not be ignored***(err2)* 
; 
plot,ax,/nodata,position=[x0,y0,x1,y1],xstyle=5,ystyle=5 
; 
;====================================== 
; plot image (x,y in real length) 
;******************************* 
; reduce file size 
; 300 dpi * 10 inches = 3000 points 
; 300 dpi * 8 inches = 2400 points 
L3000=lyplot  
L2400=lxplot 
if L3000 gt 3000 then L3000=3000 
if L2400 gt 2400 then L2400=2400 
;******************************* 
; 
; rescale the floating array data2d to an image array data2d_img 
;   which contains L2400xL3000 pixels  
;   and with data value ranges from 0-255 
; 
data2d_img=BytScl(ConGrid(data2d,L2400,L3000)) 
; 
; using TVScl to plot the image: 
; TVScl, image_array,x,y,xsize=?,ysize=?,/real_scale 
; 
; where the (x,y) is the coordinate of the lower-left corner of the image  
; w.r.t. the lower-left corner of the device (origin) in the real_scale; 
; xsize=? and ysize=? define the real size of the image to be plotted. 
; 
TVScl,data2d_img,xedge,yedge,xsize=wx,ysize=wy,/centimeters 
; 
 
;====================================== 
; plot frame again, since the frame were covered by the image. 
; 
plot,ax,/nodata,position=[x0,y0,x1,y1],$ 
  back=255,color=0,thick=2,charthick=2,$ 



  title=maintitle,xtitle='!8x/x!D0!N',ytitle='!8y/x!D0!N',$ 
  xrange=[xmin,xmax],yrange=[ymin,ymax], xstyle=1,ystyle=1 
 
 
;====================================== 
; plot color bar  
; 
; ****The following plot statement can be ignored**** 
;   If it is ignored, you should reduce the number of plots per page 
;   e.g., set !P.Multi=[0,3,1] in this example 
; 
plot,ax,/nodata,position=[x2,y2,x3,y3],xstyle=5,ystyle=5 
 
bar2d_img=BytScl(ConGrid(bar2d,L2400,L3000)) 
TVScl,bar2d_img,xbedge,yedge,xsize=wbar,ysize=wy,/centimeters 
;====================================== 
; plot frame of color bar  
;  xstyle=1+4=5 : for exact x-range and no axis will be plotted 
;;  xminor=-1  : no minor tickmark 
;;  xticks=1   : no major tickmark 
; 
; fmax0 : lower 3 points from top for 300dpi image 
; 
fmax0=fmin+(fmax-fmin)*(wy-0.0254)/wy   
plot,[0.,1.,1.,0.,0.],[fmin,fmin,fmax0,fmax0,fmin],position=[x2,y2,x3,y3],$ 
  back=255,color=0,thick=2,charthick=2,$ 
  title='',xtitle='',ytitle='',$ 
  xrange=[0.,1.],yrange=[fmin,fmax],xstyle=5,ystyle=5 
; 
;====================================== 
; yaxis=0      : plot y-axis on the left-hand-side boundary 
; yaxis=1      : plot y-axis on the right-hand-side boundary 
; yticklen=0.2 : for longer y-tickmark 
; 
axis,1,0,yaxis=1,charthick=2,yticklen=0.2,ytitle='',$ 
   xrange=[0.,1.],yrange=[fmin,fmax],ystyle=1 
;  xrange=[0.,1.],yrange=[fmin,fmax] ;this gives you incorrect color-bar (err3) 
; 
; plot additioanl subtitles using xyouts 
;x=x0+0.025 
;y=y1-0.025 
;xyouts,x,y,subtitle,size=1.5,/normal  
;====================================== 
 
device,/close 
 
exit 
 
 


